Coupled Harmonic Oscillators

Introduction

This worksheet demonstrates Maple's power in solving complex problems and shows the results graphically. The animations in the worksheet are a representation of variations in the system. The system used in this example is of a spring oscillating system.

```
> restart;
```

Mathematical Model of the System

```
> eq1:=m1*diff(x1(t),t$2)+(k1+k12)*x1(t)-k12*x2(t)=0;
```

$$eq1 := m1\left(\frac{d^2}{dt^2}\,x1(t)\right) + (k1+k12)\,x1(t) - k12\,x2(t) = 0$$

```
> eq2:=m2*diff(x2(t),t$2)-k12*x1(t)+(k12+k2)*x2(t)=0;
```

$$eq2 := m2\left(\frac{d^2}{dt^2}\,x2(t)\right) - k12\,x1(t) + (k12+k2)\,x2(t) = 0$$

```
> ics:=x1(0)=X1,x2(0)=X2,D(x1)(0)=v1,D(x2)(0)=v2;
```

$$ics := x1(0) = X1, x2(0) = X2, D(x1)(0) = v1, D(x2)(0) = v2$$

Using Maple V to Solve the System

```
> ans:=dsolve({eq1,eq2,ics},{x1(t),x2(t)}):
```

Assign the solutions for each of the blocks.

```
> block1:=subs(ans,x1(t)):  block2:=subs(ans,x2(t)):
```

Assigning Values

Assume the following values for this particular example:
m1=m2,
k1=k12=k2,
x1(0)=0,
x2(0)=0,
v1(0)=0, and
v2(0)=v.

```
> assume(k,positive); assume(m,positive);
> b1:=simplify(simplify(subs([k1=k,k2=k,k12=k,m1=m,m2=m,X1=0,X2=0,
  v1=0,v2=v],block1)));
```

$$b1 := -\frac{v\sqrt{m\sim}\left(\sqrt{3}\,\sin\left(\frac{\sqrt{3}\,\sqrt{k\sim}\,t}{\sqrt{m\sim}}\right) - 3\sin\left(\frac{\sqrt{k\sim}\,t}{\sqrt{m\sim}}\right)\right)}{6\sqrt{k\sim}}$$

```
> b2:=simplify(simplify(subs([k1=k,k2=k,k12=k,m1=m,m2=m,X1=0,X2=0,
  v1=0,v2=v],block2)));
```

$$b2 := \frac{v\sqrt{m\sim}\left(\sqrt{3}\,\sin\left(\frac{\sqrt{3}\,\sqrt{k\sim}\,t}{\sqrt{m\sim}}\right) + 3\,\sin\left(\frac{\sqrt{k\sim}\,t}{\sqrt{m\sim}}\right)\right)}{6\sqrt{k\sim}}$$

Assigning Values to the Remaining Parameters (m=1, k=1, v=1)

```
> p1:=unapply(subs([m=1,k=1,v=1],b1),t);
```

$$p1 := t \mapsto -\frac{\sqrt{3}\,\sin\left(\sqrt{3}\,t\right)}{6} + \frac{\sin(t)}{2}$$

```
> p2:=unapply(subs([m=1,k=1,v=1],b2),t);
```

$$p2 := t \mapsto \frac{\sqrt{3}\,\sin\left(\sqrt{3}\,t\right)}{6} + \frac{\sin(t)}{2}$$

```
> plot([p1(t),p2(t)],t=0..20,color=[red,blue]);
```



Note that the cycle will never repeat itself since the periods of $\sin(t)$ and $\sin\left(\sqrt{3}\,t\right)$ never match.

Procedure to Draw Springs

```
> springplot:=proc(a,b,n) local l,i,p;
    l:=(b-a)/n;
```

```
    p[0]:=plot([[a,0],[a+l,0]],color=black,thickness=2);
    for i from 1 to n-2 do                          p[i]:=plot([
[a+i*l,0],[a+i*l+l/3,1/2],[a+i*l+2*l/3,-1/2],[a+i*l+l,0]],color=
black,thickness=2);
    od;
p[n-1]:=plot([[a+(n-1)*l,0],[a+(n-1)*l+l,0]],color=black,
thickness=2);
p:=plots[display](seq(p[i],i=0..n-1));
end:
```

Animation

```
> with(plots):
```

Procedure to create animation

```
> oscillator:=proc(p1,p2,n::posint,f)
     local end1,end2,a,b,c1,c2,spring1,spring2,spring3,masses,
  i;
     end1:=plot([[0,-1],[0,1]],color=black,thickness=2);
     end2:=plot([[17,-1],[17,1]],color=black,thickness=2);
     for i from 0 to n-1 do
        a:=p1(i*f)+5; b:=p2(i*f)+11;
        c1[i]:=polygonplot([[a-1,-1],[a-1,1],[a+1,1],[a+1,-1]],
  color=red);
        c2[i]:=polygonplot([[b-1,-1],[b-1,1],[b+1,1],[b+1,-1]],
  color=blue);
        spring1[i]:=springplot(0,a-1,10);
        spring2[i]:=springplot(a+1,b-1,10);
        spring3[i]:=springplot(b+1,17,10);
        masses[i]:=display({c1[i],c2[i],
                           spring1[i],spring2[i],spring3[i],
  end1,end2});
     od:
     display([seq(masses[i],i=0..n-1)],
              insequence=true,scaling=constrained,axes=none,
  title=`Spring Oscillating System`);
  end:
> oscillator(p1,p2,40,1/2);
```

Spring Oscillating System



Solving the Equations Manually

```
> eq3:=subs([m1=1,m2=1,k1=k,k12=k,k2=k],eq1);
```

$$eq3 := \frac{d^2}{dt^2}\, x1(t) + 2\, k{\sim}\, x1(t) - k{\sim}\, x2(t) = 0$$

```
> eq4:=subs([m1=1,m2=1,k1=k,k12=k,k2=k],eq2);
```

$$eq4 := \frac{d^2}{dt^2}\, x2(t) - k{\sim}\, x1(t) + 2\, k{\sim}\, x2(t) = 0$$

Since the system is linear, assume a solution of the form
$x1(t) = a\, e^{I(\text{omega}\, t + \text{phi})}, x2(t) = b\, e^{I(\text{omega}\, t + \text{phi})}$

```
> eq5:=x1(t)=a*exp(I*(omega*t+phi)); eq6:=x2(t)=b*exp(I*(omega*t+
  phi));
```

$$eq5 := x1(t) = a\, e^{I(\omega t + \phi)}$$

$$eq6 := x2(t) = b\, e^{I(\omega t + \phi)}$$

Substitute the assumed solution into the system of equations.

```
> eq7:=simplify(subs([eq5,eq6],eq3)); eq8:=simplify(subs([eq5,
  eq6],eq4));
```

$$eq7 := \left((2\,a - b)\,k{\sim} - a\,\omega^2\right) e^{I(\omega t+\phi)} = 0$$

$$eq8 := -e^{I(\omega t+\phi)} \left((a - 2\,b)\,k{\sim} + b\,\omega^2\right) = 0$$

Divide the equations by $e^{I(\text{omega}\,t + \text{phi})}$.

```
> eq9:=simplify(eq7/exp(I*(omega*t+phi))); eq10:=simplify(eq8/exp
  (I*(omega*t+phi)));
```

$$eq9 := (2\,a - b)\,k{\sim} - a\,\omega^2 = 0$$

$$eq10 := (-a + 2\,b)\,k{\sim} - b\,\omega^2 = 0$$

Rearrange the equations for appearance.

```
> eq9:=collect(eq9,[a,b]); eq10:=collect(eq10,[a,b]);
```

$$eq9 := \left(-\omega^2 + 2\,k{\sim}\right) a - k{\sim}\,b = 0$$

$$eq10 := -k{\sim}\,a + \left(-\omega^2 + 2\,k{\sim}\right) b = 0$$

Since this system (eq9, eq10) is homogeneous, a necessary condition for a nontrivial solution to exist is that the determinant of the following matrix be zero.

```
> A:=matrix([[coeff(lhs(eq9),a),  coeff(lhs(eq9),b)],
             [coeff(lhs(eq10),a), coeff(lhs(eq10),b)]]);
```

$$A := \begin{bmatrix} -\omega^2 + 2\,k{\sim} & -k{\sim} \\ -k{\sim} & -\omega^2 + 2\,k{\sim} \end{bmatrix}$$

```
> with(linalg):
```

Calculate the determinant of A.

```
> det(A);
```

$$\omega^4 - 4\,k{\sim}\,\omega^2 + 3\,k{\sim}^2$$

At this point you could solve for det(A)=0, but in this case det(A) factors well.

```
> detA:=factor(det(A));
```

$$detA := \left(-\omega^2 + 3\,k{\sim}\right)\left(-\omega^2 + k{\sim}\right)$$

Instead of solving directly for omega, solve for omega^2.

```
> omegasquared:=solve(detA=0,omega^2);
Warning, solving for expressions other than names or functions
is not recommended.
```

$$omegasquared := k{\sim},\ 3\,k{\sim}$$

```
> omega1:=omega^2=omegasquared[1]; omega2:=omega^2=omegasquared[2]
  ;
```

$$\omega 1 := \omega^2 = k{\sim}$$

$$\omega 2 := \omega^2 = 3\,k{\sim}$$

Substitute the values for omega^2 into eq9 and eq10, and solve for a in terms of b.

```
> subs(omega1,eq9); subs(omega1,eq10);
```

$$k{\sim}\,a - k{\sim}\,b = 0$$

$$-k{\sim}\,a + k{\sim}\,b = 0$$

```
> solve(%,a);
```

$$b$$

```
> subs(omega2,eq9); subs(omega2,eq10);
```

$$-k{\sim}\,a - k{\sim}\,b = 0$$

$$-k\sim a - k\sim b = 0$$

```
> solve(%,a);
```

$$-b$$

Substitute the values found for a, b, and omega into eq5 and eq6.

```
> eq11:=convert(subs([a=a1,b=-a1,omega=sqrt(3*k)],eq5),trig);
  eq12:=convert(subs([a=a1,b=-a1,omega=sqrt(3*k)],eq6),trig);
```

$$eq11 := x1(t) = a1\left(\cos\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)+I\sin\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)\right)$$

$$eq12 := x2(t) = -a1\left(\cos\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)+I\sin\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)\right)$$

```
> eq13:=convert(subs([a=a2,b=a2,omega=sqrt(k)],eq5),trig); eq14:=
  convert(subs([a=a2,b=a2,omega=sqrt(k)],eq6),trig);
```

$$eq13 := x1(t) = a2\left(\cos\left(\sqrt{k\sim}\,t+\phi\right)+I\sin\left(\sqrt{k\sim}\,t+\phi\right)\right)$$

$$eq14 := x2(t) = a2\left(\cos\left(\sqrt{k\sim}\,t+\phi\right)+I\sin\left(\sqrt{k\sim}\,t+\phi\right)\right)$$

Remove the real parts of the answers.

```
> eq15:=lhs(eq11)=evalc(Re(rhs(eq11))+Re(rhs(eq13)));
```

$$eq15 := x1(t) = a1\cos\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)+a2\cos\left(\sqrt{k\sim}\,t+\phi\right)$$

```
> eq16:=lhs(eq12)=evalc(Re(rhs(eq12))+Re(rhs(eq14)));
```

$$eq16 := x2(t) = -a1\cos\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)+a2\cos\left(\sqrt{k\sim}\,t+\phi\right)$$

Compute the velocities (which will be required to specify initial conditions).

```
> eq17:=v1(t)=diff(rhs(eq15),t);
```

$$eq17 := v1(t) = -a1\sqrt{3}\sqrt{k\sim}\sin\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)-a2\sqrt{k\sim}\sin\left(\sqrt{k\sim}\,t+\phi\right)$$

```
> eq18:=v2(t)=diff(rhs(eq16),t);
```

$$eq18 := v2(t) = a1\sqrt{3}\sqrt{k\sim}\sin\left(\sqrt{3}\sqrt{k\sim}\,t+\phi\right)-a2\sqrt{k\sim}\sin\left(\sqrt{k\sim}\,t+\phi\right)$$

Solve for the unknown parameters using the initial conditions.

```
> parms:=solve({subs(t=0,rhs(eq15))=0,subs(t=0,rhs(eq17))=0,
        subs(t=0,rhs(eq16))=0,subs(t=0,rhs(eq18))=v},{a1,a2,phi})
  ;
```

$$parms := \left\{a1=\frac{v\sqrt{3}}{6\sqrt{k\sim}},\,a2=-\frac{v}{2\sqrt{k\sim}},\,\phi=\frac{\pi}{2}\right\},\,\left\{a1=-\frac{v\sqrt{3}}{6\sqrt{k\sim}},\,a2=\frac{v}{2\sqrt{k\sim}},\,\phi=-\frac{\pi}{2}\right\}$$

Substitute in the values of the parameters to get the final results.

```
> simplify(subs(parms[1],eq15)); simplify(subs(parms[1],eq16));
```

$$x1(t) = -\frac{v\left(\sqrt{3}\sin\left(\sqrt{3}\sqrt{k\sim}\,t\right)-3\sin\left(\sqrt{k\sim}\,t\right)\right)}{6\sqrt{k\sim}}$$

$$x2(t) = \frac{v\left(\sqrt{3}\sin\left(\sqrt{3}\sqrt{k\sim}\,t\right)+3\sin\left(\sqrt{k\sim}\,t\right)\right)}{6\sqrt{k\sim}}$$

```
> simplify(subs(parms[2],eq15)); simplify(subs(parms[2],eq16));
```

$$x1(t) = -\frac{v\left(\sqrt{3}\sin\left(\sqrt{3}\sqrt{k\sim}\,t\right)-3\sin\left(\sqrt{k\sim}\,t\right)\right)}{6\sqrt{k\sim}}$$

$$x2(t) = \frac{v\left(\sqrt{3}\sin\left(\sqrt{3}\sqrt{k\sim}\,t\right)+3\sin\left(\sqrt{k\sim}\,t\right)\right)}{6\sqrt{k\sim}}$$

Normal Modes

An interesting phenomenon occurs when the initial conditions are specified as a1=0 or a2=0 in eq15 and eq16.

```
> p3:=unapply(subs([a1=0,a2=1,k=1,phi=0],rhs(eq15)),t);
  p4:=unapply(subs([a1=0,a2=1,k=1,phi=0],rhs(eq16)),t);
```

$$p3 := t \mapsto \cos(t)$$

$$p4 := t \mapsto \cos(t)$$

```
> p5:=unapply(subs([a1=1,a2=0,k=1,phi=0],rhs(eq15)),t);
  p6:=unapply(subs([a1=1,a2=0,k=1,phi=0],rhs(eq16)),t);
```

$$p5 := t \mapsto \cos\left(\sqrt{3}\ t\right)$$

$$p6 := t \mapsto -\cos\left(\sqrt{3}\ t\right)$$

Animating the Normal Modes

Note that each of the plots in this section are animations. To view the animation, click on the plot and use the controls on the Animation toolbar.

```
> oscillator(p3,p4,10,2*Pi/10);
```

Spring Oscillating System



```
> oscillator(p5,p6,10,2*Pi/sqrt(3)/10);
```

Spring Oscillating System



Solving the System using Normal Coordinates

Consider the following change of variables.

```
> eq19:=eta1(t)=x2(t)+x1(t); eq20:=eta2(t)=x2(t)-x1(t);
```

$$eq19 := \eta 1(t) = x2(t) + x1(t)$$

$$eq20 := \eta 2(t) = x2(t) - x1(t)$$

The inverse transformation is

```
> solve({eq19,eq20},{x1(t),x2(t)});
```

$$\left\{x1(t) = \frac{\eta 1(t)}{2} - \frac{\eta 2(t)}{2}, x2(t) = \frac{\eta 2(t)}{2} + \frac{\eta 1(t)}{2}\right\}$$

```
> eq21:=%[1]; eq22:=%%[2];
```

$$eq21 := x1(t) = \frac{\eta 1(t)}{2} - \frac{\eta 2(t)}{2}$$

$$eq22 := x2(t) = \frac{\eta 2(t)}{2} + \frac{\eta 1(t)}{2}$$

```
> eq3; eq4;
```

$$\frac{d^2}{dt^2}\, x1(t) + 2\, k{\sim}\, x1(t) - k{\sim}\, x2(t) = 0$$

$$\frac{d^2}{dt^2}\, x2(t) - k{\sim}\, x1(t) + 2\, k{\sim}\, x2(t) = 0$$

Apply the transformation to normal coordinates.

```
> eq23:=simplify(subs([eq21,eq22],eq3)); eq24:=simplify(subs(
  [eq21,eq22],eq4));
```

$$eq23 := \frac{k{\sim}\, \eta 1(t)}{2} - \frac{3\, k{\sim}\, \eta 2(t)}{2} + \frac{d^2}{dt^2}\left(\frac{\eta 1(t)}{2} - \frac{\eta 2(t)}{2}\right) = 0$$

$$eq24 := \frac{k{\sim}\, \eta 1(t)}{2} + \frac{3\, k{\sim}\, \eta 2(t)}{2} + \frac{d^2}{dt^2}\left(\frac{\eta 2(t)}{2} + \frac{\eta 1(t)}{2}\right) = 0$$

Add and subtract the two equations shown above. The coupled system of differential equations is turned into an uncoupled one.

```
> eq25:=eq24+eq23; eq26:=eq24-eq23;
```

$$eq25 := k{\sim}\, \eta 1(t) + \frac{d^2}{dt^2}\, \eta 1(t) = 0$$

$$eq26 := 3\, k{\sim}\, \eta 2(t) + \frac{d^2}{dt^2}\, \eta 2(t) = 0$$

Solve the uncoupled equations.

```
> eq27:=dsolve({eq25,eta1(0)=0,D(eta1)(0)=v},{eta1(t)});
```

$$eq27 := \eta 1(t) = \frac{v \sin\left(\sqrt{k{\sim}}\, t\right)}{\sqrt{k{\sim}}}$$

```
> eq28:=dsolve({eq26,eta2(0)=0,D(eta2)(0)=v},{eta2(t)});
```

$$eq28 := \eta 2(t) = \frac{v \sqrt{3}\, \sin\left(\sqrt{3}\, \sqrt{k{\sim}}\, t\right)}{3\sqrt{k{\sim}}}$$

Apply the inverse transformation to arrive at the same answer as before.

```
> simplify(subs([eq27,eq28],eq21)); simplify(subs([eq27,eq28],
  eq22));
```

$$x1(t) = -\frac{v\left(\sqrt{3}\, \sin\left(\sqrt{3}\, \sqrt{k{\sim}}\, t\right) - 3 \sin\left(\sqrt{k{\sim}}\, t\right)\right)}{6\sqrt{k{\sim}}}$$

$$x2(t) = \frac{v\left(\sqrt{3}\, \sin\left(\sqrt{3}\, \sqrt{k{\sim}}\, t\right) + 3 \sin\left(\sqrt{k{\sim}}\, t\right)\right)}{6\sqrt{k{\sim}}}$$